

Neue Möglichkeiten bei der Hochfrequenzberechnung

Ausgabe: 1 / 2003

Problem:

Komplexe Bauteile sind häufig nicht oder nur mit sehr großem Aufwand mit Hexaeder-Elementen vernetzbar. Hier hat sich bewährt, stattdessen Tetraeder-Elemente zu verwenden. Allerdings ist die Elementanzahl bei Tetraedervernetzung deutlich höher als bei Hexaedervernetzung. Hier bietet es sich an, einfache Körper mit Hexaedern, komplexere Bauteile mit Tetraeder zu vernetzen. Ab der Version 7.0 wird nun ein Hexaeder-Tetraeder-Übergang durch Pyramiden-Elemente unterstützt.

Außerdem kann ANSYS nun eine adaptive Netzverfeinerung an Stellen mit einem hohen lokalen Diskretisierungsfehler durchführen. Dies ist allerdings nur für reine Tetraeder-Netze möglich.

Erläuterung:






Bisher wurde bei den Hochfrequenzelementen HF120 nur die Hexaederform berücksichtigt. Ab ANSYS 7.0 ist es möglich, wie bei den strukturmechanischen Elementen eine gemischte Vernetzung aus Hexaeder- und Tetraederelementen vorzunehmen. Der Übergang wird dann durch Pyramidenelemente gebildet. Die Tetraedervernetzung wird durch MSHAPE,1,3D erzwungen. Wenn schon ein benachbartes Volumen hexaedervernetzt ist, wird automatisch der Hexaeder-Tetraeder-Übergang durchgeführt. Anschließend müssen mit TCHG,120,119 die degenerierten HF120-Tetraeder in HF119-Elemente gewandelt werden. Wird dies nicht durchgeführt, kann ANSYS die Berechnung nicht starten.

Bei reinen Tetraedernetzen kann eine adaptive Netzverfeinerung mit HFEREFINE,FAKTOR vorgenommen werden. Als Kriterium dient der lokale Diskretisierungsfehler; es werden alle Elemente, deren Fehler $>$ FAKTOR * durchschnittlicher Fehler ist, verfeinert. Es können Faktoren zwischen 1.0 und 3.0 verwendet werden, wobei Faktor 1 maximale Verfeinerung bedeutet. Der Befehl kann nur verwendet werden, wenn ausschließlich Elementtypen HF119 definiert wurden. Sind noch Elementtypen mit HF120 definiert (ET,...,120), erscheint eine Fehlermeldung.

Beispiel:

Das Beispiel zeigt einen Hohlraumresonator der Abmessungen 15mm x 10mm x 9mm. Dieser wird über eine Zylinderschraube verstimmt. Der Bereich des Resonators, in den die Schraube ragt, wird mit Tetraedern vernetzt, der andere Bereich und die Schraube mit Hexaedern. Die Wände von Resonator und Schraube werden als ideal leitend angenommen (D,ALL,AX,0), was durch die verschwindend geringen Eindringtiefen bei den auftretenden Frequenzen zulässig ist. Von Interesse ist die erste Resonanzfrequenz, die durch eine Modalanalyse gefunden wird. Im ersten Schritt wird diese ohne Berücksichtigung der Schraube bestimmt, im zweiten Lastschritt die Frequenz des verstimmten Systems.

Durch Umschalten der Variablen m_ctrl im Eingabestrom auf 2 wird statt des Hexaeder/Tetraeder-Übergangs die gesamte Struktur tet-ernetzt. Auch hier werden die Resonanzen für das System ohne und mit Schraube bestimmt. Anschließend wird eine adaptive Netzverfeinerung (HFEREFINE,3) durchgeführt und erneut eine Modalanalyse vorgenommen.



Neue Möglichkeiten bei der Hochfrequenzberechnung

Ausgabe: 1 / 2003

ANSYS Eingabesatz (ANSYS 7.0):

```
fini
/cle

m_ctrl=1      ! 1: hex/tet
               ! 2: nur tet mit hferefine

/prep7
bloc,0,15e-3,0,10e-3,0,9e-3
lwplan,,5,0.5 ! in 2 Volumen zerschneiden
vsbw,all
wpcsys,-1     ! Schraube
cyl4,.5*7.5e-3,5e-3,1e-3,,2e-3
vovl,all

mp,murx,1,1
mp,perx,1,1

lsel,s,,2,4,2 ! Netzdichte
               ! Schraubenflächen

lsel,a,,5,7,2
lsel,a,,25,28
lesize,all,,2
esize,10e-3/10

*if, m_ctrl,eq,1,then
et,1,120,1    ! HF Hexaeder
et,2,119,1    ! HF Tetraeder
type,1        ! hex/tet mesh
vsweep,1     ! Hexaeder Vernetzung
vmesh,3
mshape,1,3d! Tetraeder Vernetzung
vmesh,4      ! mit automatischer
               ! Pyramidengenerierung

tchg,120,119 ! 120er Tets in 119er
               ! konvertieren

*else
et,1,119,1    ! HF Tetraeder
mshape,1,3d! Tetraeder Vernetzung
vmesh,all
*endif
```

```
/solu
nset,s,ext
d,all,ax,0    ! leitfähiger Rand
alls

antype,modal
modopt,lanb,1,5e9,20e9 ! nur 1.
                       ! Resonanz v. Interesse
mxpand,,,,yes
solve         ! 1.Loadstep ohne Schraube

vsel,s,,1    ! Schr.flächen als id. leitend
eslv,s
nset,s,ext
d,all,ax,0
allsel
solve        ! 2. Loadstep mit Schraube

/post1
set,1,1      ! Resonanz ohne Schraube
plns,h,sum   ! Analytisch bei
               ! 18.028GHz
set,2,1      ! Resonanz durch Schraube
plns,h,sum   ! verstimmt

*if, m_ctrl,ne,1,then
hferefine,3  ! adapt. Netzverfeinerung
/solu        ! nur bei reinem Tet-Netz
nset,s,ext   ! möglich
d,all,ax,0   ! leitfähiger Rand
alls
vsel,s,,1    ! Schr.flächen als id. leitend
eslv,s
nset,s,ext
d,all,ax,0
allsel
solve        ! Neuer LS mit Schraube
*endif

/post1
plns,h,sum
```